

# SWI-Prolog SGML/XML parser

*Version 1.0.6, July 19 2000*

*Jan Wielemaker*

SWI,  
University of Amsterdam  
The Netherlands  
**E-mail:** `jan@swi.psy.uva.nl`

## **Abstract**

Markup languages are an increasingly important method for data-representation and exchange. This article documents the package **sgml2pl**, a foreign library for SWI-Prolog to parse SGML and XML documents, returning information on both the document and the document's DTD. The parser is designed to be small, fast and flexible.

# Contents

<b>1</b>	<b>Introduction</b>	<b>1</b>
<b>2</b>	<b>Bluffers Guide</b>	<b>2</b>
2.1	‘Goodies’ Predicates . . . . .	3
<b>3</b>	<b>Predicate Reference</b>	<b>4</b>
3.1	Loading Structured Documents . . . . .	4
3.2	Handling white-space . . . . .	5
3.3	XML documents . . . . .	6
3.3.1	XML Namespaces . . . . .	7
3.4	DTD-Handling . . . . .	7
3.4.1	The DOCTYPE declaration . . . . .	10
3.5	Extracting a DTD . . . . .	10
3.6	Parsing Primitives . . . . .	10
3.6.1	Partial Parsing . . . . .	13
<b>4</b>	<b>Processing Indexed Files</b>	<b>14</b>
<b>5</b>	<b>External entities</b>	<b>15</b>
<b>6</b>	<b>Missing functionality</b>	<b>16</b>
<b>7</b>	<b>Installation</b>	<b>16</b>
7.1	Unix systems . . . . .	16
7.2	Acknowledgements . . . . .	17
<b>A</b>	<b>Summary of Predicates</b>	<b>18</b>

pagebreak

## 1 Introduction

Markup languages have recently regained popularity for two reasons. One is document exchange, which is largely based on HTML, an instance of SGML and the other is for data-exchange between programs, which is often based on XML, which can be considered simplified and rationalised version of SGML.

James Clark’s SP parser is a flexible SGML and XML parser. Unfortunately it has some drawbacks. It is very big, not very fast, cannot work under event-driven input and is generally hard to program beyond the scope of the well designed generic interface. The generic interface however does not provide access to the DTD, does not allow for flexible handling of input or parsing the DTD independently of a document instance.

The parser described in this document is small (less than 50 Kbytes executable on a Pentium), fast (between 2 and 5 times faster than SP), provides access to the DTD and flexible input handling.

The document output is equal to the output produced by *xml2pl*, an SP interface to SWI-Prolog written by Anjo Anjewierden.

## 2 Bluffers Guide

This package allows you to parse SGML, XML and HTML data into a Prolog data structure. The high-level interface defined in **sgml** provides access at the file-level, while the low-level interface defined in the foreign module works with Prolog streams. Please use the source of **sgml.pl** as a starting point for dealing with data from other sources than files, such as SWI-Prolog resources, network-sockets, character strings, etc. In the first example below loads an HTML file.

```
<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 3.2//EN">
```

```
<html>
<head>
<title>Demo</title>
</head>
<body>
```

```
<h1 align=center>This is a demo</title>
```

```
<p>Paragraphs in HTML need not be closed.
```

```
<p>This is called 'omitted-tag' handling.
</body>
</html>
```

```
?- load_html_file('test.html', Term),
   pretty_print(Term).
```

```
[ element(html,
  [],
  [ element(head,
    [],
    [ element(title,
      [],
      [ 'Demo'
      ])
    ],
    element(body,
      [],
      [ '\n',
        element(h1,
          [ align = center
```

```

],
[ 'This is a demo'
]),
'\n\n',
element(p,
[],
[ 'Paragraphs in HTML need not be closed.\n'
]),
element(p,
[],
[ 'This is called 'omitted-tag\' handling.'
])
])
])
].

```

The document is represented as a list, each element being an atom to represent CDATA or a term `element(Name, Attributes, Content)`. Entities (e.g. `&lt;`;) are returned as part of CDATA, unless they cannot be represented. See `load_sgml_file/2` for details.

## 2.1 ‘Goodies’ Predicates

These predicates are for basic usage of the library, converting entire and self-contained files in one of the three supported XML dialects into a structured term. They are all expressed in `load_structure/3`.

**load\_sgml\_file(+File, -ListOfContent)**

Same as `load_structure(File, ListOfContent, [dialect(sgml)])`.

**load\_xml\_file(+File, -ListOfContent)**

Same as `load_structure(File, ListOfContent, [dialect(xml)])`.

**load\_html\_file(+File, -Content)**

Load *File* and parse as HTML. Implemented as:

```

load_html_file(File, Term) :-
    dtd(html, DTD),
    load_structure(File, Term,
        [ dtd(DTD),
          dialect(sgml)
        ]).

```

## 3 Predicate Reference

### 3.1 Loading Structured Documents

SGML or XML files are loaded through the common predicate **load\_structure/3**. This is a predicate with many options. For simplicity a number of commonly used shorthands are provided: **load\_sgml\_file/2**, **load\_xml\_file/2**, and **load\_html\_file/2**.

**load\_structure**(*+File*, *-ListOfContent*, *+Options*)

Load the XML file *File* and return the resulting structure in *ListOfContent*. *Options* is a list of options controlling the conversion process.

A proper XML document contains only a single toplevel element whose name matches the document type. Nevertheless, a list is returned for consistency with the representation of element content. The *ListOfContent* consists of three types:

#### *Atom*

Atoms are used to represent CDATA. Note this is possible in SWI-Prolog, as there is no length-limit on atoms and atom garbage collection is provided.

**element**(*Name*, *ListAttributes*, *ListOfContent*)

*Name* is the name of the element. Using SGML, which is case-insensitive, all element names are returned as lowercase atoms.

*ListOfAttributes* is a list of *Name=Value* pairs for attributes that appeared in the source. No information is returned on other attributes, such as **fixed** or **default** attributes. See **dtd\_property/2** for accessing the DTD for this information. Attributes of type CDATA are returned literal. Attributes of type NUMBER are returned as a Prolog integer. Multi-valued attributes (NAMES, etc.) are returned as a list of atoms. For the type NUMBERS, a list of Prolog integers is returned for any element of the list that can be converted into an integer.

*ListOfContent* defines the content for the element.

**entity**(*Code*)

If a character-entity (e.g. `&#913;`) is encountered that cannot be represented in the Prolog character set, this term is returned, representing the referred character code.

**entity**(*Name*)

If an entity refers to a character-entity holding a single character, but this character cannot be represented in the Prolog character set, this term is returned. For example, the HTML input text `&Alpha; &lt; &Beta;` is returned as below. Please note that neither in XML nor SGML entities are case-insensitive.

```
[ entity('Alpha'), ' < ', entity('Beta') ]
```

This is a special case of **entity(Code)**, intended to handle special symbols by their name rather than character code.

**sdata**(*Text*)

If an entity with declared content-type SDATA is encountered, this term is returned holding the data in *Text*.

**ndata**(*Text*)

If an entity with declared content-type **CDATA** is encountered, this term is returned holding the data in *Text*.

**pi**(*Text*)

If a processing instruction is encountered (`<?...?>`), *Text* holds the text of the processing instruction. Please note that the `<?xml ...?>` instruction is handled internally.

The *Options* list controls the conversion process. Currently defined options are:

**dtd**(*?DTD*)

Reference to a DTD object. If specified, the `<!DOCTYPE ...>` declaration is ignored and the document is parsed and validated against the provided DTD. If provided as a variable, the implicitly created DTD is returned. See section 3.5.

**dialect**(*+Dialect*)

Specify the parsing dialect. Supported are **sgml** (default), **xml** and **xmlns**. See section 3.3 for details on the differences.

**space**(*+SpaceMode*)

Sets the ‘space-handling-mode’ for the initial environment. This mode is inherited by the other environments and subject to the XML reserved tag **xml:space**. See section 3.2.

**file**(*+Name*)

Sets the name of the file on which errors are reported. Sets the linenumber to 1.

**line**(*+Line*)

Sets the starting line-number for reporting errors.

**max\_errors**(*+Max*)

Sets the maximum number of errors. If this number is reached, an exception of the format below is raised. The default is 50.

`error(limit_exceeded(max_errors, Max), _)`

## 3.2 Handling white-space

SGPL2PL has four modes for handling white-space. The initial mode can be switched using the **space**(*SpaceMode*) option to **load\_structure/3** and **set\_sgml\_parser/2**. In XML mode, the mode is further controlled by the **xml:space** attribute, which may be specified both in the DTD as in the document. The defined modes are:

**space**(*sgml*)

In SGML, newlines at the start and end of an element are removed.<sup>1</sup> This is the default mode for the SGML dialect.

---

<sup>1</sup>In addition, newlines at the end of lines containing only markup should be deleted. This is not yet implemented.

#### **space**(*preserve*)

White space is passed literally to the application. This is the default mode. This mode leaves all white space handling to the application. This is the default mode for the XML dialect.

#### **space**(*default*)

In addition to **sgml** space-mode, all consecutive white-space is reduced to a single space-character. This mode canonises all white space.

#### **space**(*remove*)

In addition to **default**, all leading and trailing white-space is removed from **CDATA** objects. If, as a result, if the **CDATA** becomes empty, nothing is passed to the application. This mode is especially handy for processing ‘data-oriented’ documents, such as RDF. It is not suitable for normal text documents. Consider the HTML statement below. When processed in this mode, the spaces between the three modified words are lost. This mode is, unlike the two others, not part of the XML standard.

Consider adjacent `<b>bold</b> <ul>and</ul> <it>italic</it>` words.

### 3.3 XML documents

The parser can operate in two modes: **sgml** mode and **xml** mode, as defined by the **dialect**(*Dialect*) option. Regardless of this option, if the first line of the document reads as below, the parser is switched automatically into XML mode.

```
<?xml ... ?>
```

Currently switching to XML mode implies:

- *XML empty elements*  
The construct `<element [attribute...] />` is recognised as an empty element.
- *Predefined entities*  
The following entities are predefined: **lt** (<), **gt** (>), **amp** (&), **apos** (') and **quot** (").
- *Case sensitivity*  
In XML mode, names are treated case-sensitive, except for the DTD reserved names (i.e. **ELEMENT**, etc.).
- *Character classes*  
In XML mode, underscores (`_`) and colon (`:`) are allowed in names.
- *White-space handling*  
White space mode is set to **preserve**. In addition to setting white-space handling at the toplevel the XML reserved attribute **xml:space** is honoured. It may appear both in the document as the DTD. The **remove** extension is honoured as **xml:space** value. For example, the DTD statement below ensures that the **pre** element preserves space, regardless of the default processing mode.

```
<!ATTLIST pre xml:space nmtoken #fixed preserve>
```

### 3.3.1 XML Namespaces

Using the *dialect* `xmlns`, the parser will interpret XML namespaces. In this case, the names of elements are returned as a term of the format

*URL:LocalName*

If an identifier has no namespace and there is no default namespace it is returned as a simple atom. If an identifier has a namespace but this namespace is undeclared, the namespace name rather than the related URL is returned.

Attributes declaring namespaces (`xmlns:ns=url`) are reported as if `xmlns` is not a defined resource.

In many cases, getting attribute-names as *url:name* is not desirable. Such terms are hard to unified and sometimes multiple URLs may be mapped to the same identifier. This may happen due to poor version management, poor standardisation or because the application doesn't care too much about versions. This package defines two call-backs that can be set using `set_sgml_parser/2` to deal with this problem.

the call-back `xmlns` is called as XML namespaces are pushed on the environment. It can be used to extend a canonical mapping for later use by the `urlns` call-back. The following illustrates this behaviour. Any namespace containing `rdf-syntax` in its URL or that is used as `rdf` namespace is canonised to `rdf`. This implies that any attribute and element name from the RDF namespace appears as *rdf:name*.

```
:- dynamic
    xmlns/3.

on_xmlns(rdf, URL, _Parser) :- !,
    asserta(xmlns(URL, rdf, _)).
on_xmlns(_, URL, _Parser) :-
    sub_atom(URL, _, _, _, rdf-syntax), !,
    asserta(xmlns(URL, rdf, _)).

load_rdf_xml(File, Term) :-
    load_structure(File, Term,
        [ dialect(xmlns),
          call(xmlns, on_xmlns),
          call(urlns, xmlns)
        ]).
```

### 3.4 DTD-Handling

The DTD (**D**ocument **T**ype **D**efinition) is a separate entity in `sgml2pl`, that can be created, freed, defined and inspected. Like the parser itself, it is filled by opening it as a Prolog output stream and sending data to it. This section summarises the predicates for handling the DTD.



**new\_dtd**(+DocType, -DTD)

Creates an empty DTD for the named *DocType*. The returned DTD-reference is an opaque term that can be used in the other predicates of this package.

**free\_dtd**(+DTD)

Deallocate all resources associated to the DTD. Further use of *DTD* is invalid.

**load\_dtd**(+DTD, +File)

Define the DTD by loading the SGML-DTD file *File*. This predicate is defined using the low-level **open\_dtd/3** predicate:

```
load_dtd(DTD, DtdFile) :-
    open_dtd(DTD, [], DtdOut),
    open(DtdFile, read, DtdIn),
    copy_stream_data(DtdIn, DtdOut),
    close(DtdIn),
    close(DtdOut).
```

**open\_dtd**(+DTD, +Options, -OutputStream)

Open a DTD as an output stream. The option-list is currently empty. See **load\_dtd/2** for an example.

**dtd**(+DocType, -DTD)

Find the DTD representing the indicated *doctype*. This predicate uses a cache of DTD objects. If a doctype has no associated dtd, it searches for a file using the file search path **dtd** using the call:

```
...,
absolute_file_name(dtd(Type),
    [ extensions([dtd]),
      access(read)
    ], DtdFile),
...
```

**dtd\_property**(+DTD, ?Property)

This predicate is used to examine the content of a DTD. Property is one of:

**doctype**(DocType)

An atom representing the document-type defined by this DTD.

**elements**(ListOfElements)

A list of atoms representing the names of the elements in this DTD.

**element**(Name, Omit, Content)

The DTD contains an element with the given name. *Omit* is a term of the format **omit**(OmitOpen, OmitClose), where both arguments are booleans (**true** or **false** representing whether the open- or close-tag may be omitted). *Content* is the content-model of the element represented as a Prolog term. This term takes the following form:

**empty**  
The element has no content.

**cdata**  
The element contains non-parsed character data. All data upto the matching end-tag is included in the data (*declared content*). *itemelement/any/* The element may contain any number of any element from the DTD in any order.

**#pcdata**  
The element contains parsed character data

***element***  
An element with this name.

**\*(*SubModel*)**  
0 or more appearances.

**?(*SubModel*)**  
0 or one appearance.

**+(*SubModel*)**  
1 or more appearances.

**,(*SubModel1*, *SubModel2*)**  
*SubModel1* followed by *SubModel2*.

**&(*SubModel1*, *SubModel2*)**  
*SubModel1* and *SubModel2* in any order.

**|(*SubModel1*, *SubModel2*)**  
*SubModel1* or *SubModel2*.

**attributes(*Element*, *ListOfAttributes*)**  
*ListOfAttributes* is a list of atoms representing the attributes of the element *Element*.

**attribute(*Element*, *Attribute*, *Type*, *Default*)**  
Query an element. *Type* is one of *cdata*, *entity*, *id*, *idref*, *name*, *nmtoken*, *notation*, *number* or *nutoken*. For DTD types that allow for a list, the notation *list(Type)* is used. Finally, the DTD construct (*a|b|...*) is mapped to the term *nameof(ListOfValues)*.

**entities(*ListOfEntities*)**  
*ListOfEntities* is a list of atoms representing the names of the defined entities.

**entity(*Name*, *Value*)**  
*Name* is the name of an entity with given value. *Value* is one of

***Atom***  
If the value is atomic, it represents the literal value of the entity.

**system(*Url*)**  
*Url* is the URL of the system external entity.

**public(*Id*, *Url*)**  
For external public entities, *Id* is the identifier. If an URL is provided this is returned in *Url*. Otherwise this argument is unbound.

**notations(*ListOfNotations*)**  
Returns a list holding the names of all NOTATION declarations.

**notation(*Name*, *File*)**  
Yields the declared file for from a NOTATION declaration.

### 3.4.1 The DOCTYPE declaration

As this parser allows for processing partial documents and process the DTD separately, the DOCTYPE declaration plays a special role.

If a document has no DOCTYPE declaration, the parser returns a list holding all elements and CDATA found. If the document has a DOCTYPE declaration, the parser will open the element defined in the DOCTYPE as soon as the first real data is encountered.

## 3.5 Extracting a DTD

Some documents have no DTD. One of the neat facilities of this library is that it builds a DTD while parsing a document with an *implicit* DTD. The resulting DTD contains all elements encountered in the document. For each element the content model is a disjunction of elements and possibly #PCDATA that can be repeated. Thus, if in element **x** we found element **y** and CDATA, the model is:

```
<!ELEMENT x - - (y|#PCDATA)*>
```

Any encountered attribute is added to the attribute list with the type CDATA and default #IMPLIED.

The example below extracts the elements used in an unknown XML document.

```
elements_in_xml_document(File, Elements) :-
    load_structure(File, _,
        [ dialect(xml),
          dtd(DTD)
        ]),
    dtd_property(DTD, elements(Elements)),
    free_dtd(DTD).
```

## 3.6 Parsing Primitives

**new\_sgml\_parser**(-Parser, +Options)

Creates a new parser. A parser can be used one or multiple times for parsing documents or parts thereof. It may be bound to a DTD or the DTD may be left implicit, in which case it is created from the document prologue or parsing is performed without a DTD. Options:

**dtd**(?DTD)

If specified with an initialised DTD, this DTD is used for parsing the document, regardless of the document prologue. If specified using as a variable, a reference to the created DTD is returned. This DTD may be created from the document prologue or build implicitly from the document's content.

**free\_sgml\_parser**(*+Parser*)

Destroy all resources related to the parser. This does not destroy the DTD if the parser was created using the **dtd**(DTD) option.

**set\_sgml\_parser**(*+Parser, +Option*)

Sets attributes to the parser. Currently defined attributes:

**file**(*File*)

Sets the file for reporting errors and warnings. Sets the line to 1.

**line**(*Line*)

Sets the current line. Useful if the stream is not at the start of the (file) object for generating proper line-numbers.

**dialect**(*Dialect*)

Set the markup dialect. Known dialects:

**sgml**

The default dialect is to process as SGML. This implies markup is case-insensitive and standard SGML abbreviation is allowed (abbreviated attributes and omitted tags).

**xml**

This dialect is selected automatically if the processing instruction `<?xml ...>` is encountered. See section 3.3 for details.

**xmlns**

Process file as XML file with namespace support. See section 3.3.1 for details.

**get\_sgml\_parser**(*+Parser, -Option*)

Retrieve information on the current status of the parser. Notably useful if the parser is used in the call-back mode. Currently defined options:

**file**(*-File*)

Current file-name. Note that this may be different from the provided file if an external entity is being loaded.

**charpos**(*-CharPos*)

Offset from where the parser started its processing in the file-object. See section 4.

**source**(*-Stream*)

Prolog stream being processed. May be used in the **on\_begin**, etc. callbacks from **sgml\_parse/2**.

**dialect**(*-Dialect*)

Return the current dialect used by the parser (**sgml**, **xml** or **xmlns**).

**sgml\_parse**(*+Parser, +Options*)

Parse an XML file. The parser can operate in two input and two output modes. Output is either a structured term as described with **load\_structure/2** or call-backs on predefined events. The first is especially suitable for manipulating not-too-large documents, while the latter provides a primitive means for handling very large documents.

Input is either a stream or an goal that pushes characters into the parser. A full description of the option-list is below.

**document(+Term)**

A variable that will be unified with a list describing the content of the document (see **load\_structure/2**).

**source(+Stream)**

An input stream that is read. Either this option or the **goal(Goal)** option must be provided.

**goal(+Goal)**

*Goal* is a callable term. The predicate **sgml\_parse/2** opens an output stream to the parser and invokes **call(Goal, Stream)**, where *Goal* should write the data to be parsed to *Stream*. This option is not compatible to **parse(element)**. This option can be used for example to parse a Prolog atom:

```

parse_atom(Atom, Term) :-
    new_sgml_parser(Parser, []),
    sgml_parse(Parser,
        [ document(Term),
          goal(provide_atom(Atom))
        ]),
    free_sgml_parser(Parser).

provide_atom(Atom, ParserStream) :-
    write(ParserStream, Atom).

```

For example:

```

?- parse_atom('<h1>hello world</title>', X).

X = [element(h1, [], ['hello world'])]

```

**parse(Unit)**

If **file** (default), parse everything upto the end of the input. If **element**, the parser will stop after reading the first element. Using **source(Stream)**, this implies reading is stopped as soon as the element is complete, and another call may be issued on the same stream to read the next element. Using **goal(Goal)** as input, the stream reports an I/O error after completing the first element. This exception destroys the built **document(Term)**, making this option useless using 'Goal' driven input. The value **content** may be used in a call-back from **call(on\_begin, Pred)** to parse individual elements after validating their headers.

**max\_errors(+MaxErrors)**

Set the maximum number of errors. If this number is exceeded further writes to the stream will yield an I/O error exception. Printing of errors is suppressed after reaching this value. The default is 100.

**call(+Event, :PredicateName)**

Issue call-backs on the specified events. *PredicateName* is the name of the predicate to call on this event, possibly prefixed with a module identifier. The defined events are:

**begin**

An open-tag has been parsed. The named handler is called with three arguments: *Handler(+Tag, +Attributes, +Parser)*.

**end**

A close-tag has been parsed. The named handler is called with two arguments: *Handler(+Tag, +Parser)*.

**cdata**

CDATA has been parsed. The named handler is called with two arguments: *Handler(+CDATA, +Parser)*, where CDATA is an atom representing the data.

**entity**

An entity that cannot be represented as CDATA has been parsed. The named handler is called with two arguments: *Handler(+NameOrCode, +Parser)*.

**pi**

A processing instruction has been parsed. The named handler is called with two arguments: *Handler(+Text, +Parser)*, where *Text* is the text of the processing instruction.

**xmlns**

When parsing an in **xmlns** mode, a new namespace declaration is pushed on the environment. The named handler is called with three arguments: *Handler(+NameSpace, +URL, +Parser)*. See section 3.3.1 for details.

**urlns**

When parsing an in **xmlns** mode, this predicate can be used to map a url into either a canonical URL for this namespace or another internal identifier. See section 3.3.1 for details.

### 3.6.1 Partial Parsing

In some cases, part of a document needs to be parsed. One option is to use **load\_structure/2** or one of its variations and extract the desired elements from the returned structure. This is a clean solution, especially on small and medium-sized documents. It however is unsuitable for parsing really big documents. Such documents can only be handled with the call-back output interface realised by the `call( Event, Action)` option of **sgml\_parse/2**. Event-driven processing is not very natural in Prolog.

The SGML2PL library allows for a mixed approach. Consider the case where we want to process all descriptions from RDF elements in a document. The code below calls **process\_rdf\_description/1** on any element that is directly inside an RDF element.

```
:- dynamic
    in_rdf/0.

load_rdf(File) :-
    retractall(in_rdf),
    open(File, read, In),
    new_sgml_parser(Parser, []),
    set_sgml_parser(Parser, file(File)),
```

```

    set_sgml_parser(Parser, dialect(xml)),
    sgml_parse(Parser,
        [ source(In),
          call(begin, on_begin),
          call(end, on_end)
        ]),
    close(In).

on_end('RDF', _) :-
    retractall(in_rdf).

on_begin('RDF', _, _) :-
    assert(in_rdf).
on_begin(Tag, Attr, Parser) :-
    in_rdf, !,
    sgml_parse(Parser,
        [ document(Content),
          parse(content)
        ]),
    process_rdf_description(element(Tag, Attr, Content)).

```

## 4 Processing Indexed Files

In some cases applications which to process small portions of large SGML, XML or RDF files. For example, the *OpenDirectory* project by Netscape has produced a 90MB RDF file representing the main index. The parser described here can process this document as a unit, but loading takes 85 seconds on a Pentium-II 450 and the resulting term requires about 70MB global stack. One option is to process the entire document and output it as a Prolog fact-based of RDF tripplets, but in many cases this is undesirable. Another example is a large SGML file containing online documentation. The application normally wishes to provide only small portions at a time to the user. Loading the entire document into memory is then undesirable.

Using the `parse(element)` option, we open a file, seek (using `seek/4`) to the position of the element and read the desired element.

The index can be built using the call-back interface of `sgml_parse/2`. For example, the following code makes an index of the `structure.rdf` file of the OpenDirectory project:

```

:- dynamic
    location/3.                                % Id, File, Offset

rdf_index(File) :-
    retractall(location(_, _)),
    open(File, read, In, [type(binary)]),
    new_sgml_parser(Parser, []),
    set_sgml_parser(Parser, file(File)),
    set_sgml_parser(Parser, dialect(xml)),

```

```

sgml_parse(Parser,
            [ source(In),
              call(begin, index_on_begin)
            ]),
close(In).

index_on_begin(_Element, Attributes, Parser) :-
    memberchk('r:id'=Id, Attributes),
    get_sgml_parser(Parser, charpos(Offset)),
    get_sgml_parser(Parser, file(File)),
    assert(location(Id, File, Offset)).

```

The following code extracts the RDF element with required id:

```

rdf_element(Id, Term) :-
    location(Id, File, Offset),
    load_structure(File, Term,
                  [ dialect(xml),
                    offset(Offset),
                    parse(element)
                  ])).

```

## 5 External entities

While processing an SGML document the document may refer to external data. This occurs in three places: external parameter entities, normal external entities and the DOCTYPE declaration. The current version of these tools deal rather primitively with external data. External entities can only be loaded from a file and the mapping between the entity names and the file is done using a *catalog* file in a format compatible to that used by James Clark's SP Parser.

Catalog files can be specified using two primitives: the predicate **sgml\_register\_catalog\_file/2** or the environment variable **SGML\_CATALOG\_FILES** (compatible to the SP package).

**sgml\_register\_catalog\_file**(*+File*, *+Location*)

Register the indicated *File* as a catalog file. *Location* is either **start** or **end** and defines whether the catalog is considered first or last. This predicate has no effect if *File* is already part of the catalog.

If no files are registered using this predicate, the first query on the catalog examines **SGML\_CATALOG\_FILES** and fills the catalog with all files in this path.

Two types of lines are used by this package.

```

DOCTYPE doctype file
PUBLIC  " Id " file

```



The specified *file* path is taken relative to the location of the catalog file. For the DOCTYPE declaration, **sgml2pl** first makes an attempt to resolve the **SYSTEM** or **PUBLIC** identifier. If this fails it tries to resolve the *doctype* using the provided catalog files.

In the future we will design a call-back mechanism for locating and processing external entities, so Prolog-based file-location and Prolog resources can be used to store external entities.

## 6 Missing functionality

The current parser is rather limited. Though suitable to deal with many serious documents, it also fails a number of less-used features in SGML and XML. Known missing SGML features include

- *NOTATION on entities*  
Though notation is fully parsed, notation attributes on external entity declarations are not handed to the user.
- *SHORTTAG*  
The SGML SHORTTAG syntax is only partially implemented. Currently, `<tag/content/` is a valid abbreviation for `<tag>content</tag>`, which can also be written as `<tag>content</>`.
- *Declared content c/RCDATA/*  
The *declared* type `c/RCDATA/` is not supported.
- *SGML declaration*  
The ‘SGML declaration’ is fixed, though most of the parameters are handled through indirections in the implementation.

In XML mode the parser recognises SGML constructs that are not allowed in XML. Also various extensions of XML over SGML are not yet realised.

## 7 Installation

### 7.1 Unix systems

Installation on Unix system uses the commonly found **configure**, **make** and **make install** sequence. SWI-Prolog should be installed before building this package. If SWI-Prolog is not installed as **pl**, the environment variable **PL** must be set to the name of the SWI-Prolog executable. Installation is now accomplished using:

```
% ./configure
% make
% make install
```

This installs the foreign libraries in **\$PLBASE/lib/\$PLARCH** and the Prolog library files in **\$PLBASE/library**, where **\$PLBASE** refers to the SWI-Prolog ‘home-directory’.

## 7.2 Acknowledgements

The Prolog representation for parsed documents is based on the SWI-Prolog interface to SP by Anjo Anjewierden.

Richard O’Keefe pointed out a number of mistakes in a earlier version of this parser.

## A Summary of Predicates

<b>dtd/2</b>	Find or build a DTD for a document type
<b>dtd_property/2</b>	Query elements, entities and attributes in a DTD
<b>free_dtd/1</b>	Free a DTD object
<b>free_sgml_parser/1</b>	Destroy a parser
<b>get_sgml_parser/2</b>	Get parser options
<b>load_dtd/2</b>	Read DTD information from a file
<b>load_html_file/2</b>	Parse HTML file into Prolog term
<b>load_sgml_file/2</b>	Parse SGML file into Prolog term
<b>load_structure/3</b>	Parse XML/SGML/HTML file into Prolog term
<b>load_xml_file/2</b>	Parse XML file into Prolog term
<b>new_dtd/2</b>	Create a DTD object
<b>new_sgml_parser/2</b>	Create a new parser
<b>open_dtd/3</b>	Open a DTD object as an output stream
<b>set_sgml_parser/2</b>	Set parser options (dialect, source, etc.)
<b>sgml_parse/2</b>	Parse the input
<b>sgml_register_catalog_file/2</b>	Register a catalog file

## Index

#pcdata, 9  
cdata, 9  
empty, 9  
sgml, 11  
xmlns, 11  
xml, 11  
:, 7  
CDATA, 3, 4, 6, 10  
DOCTYPE, 15, 16  
NAMES, 4  
NDATA, 5  
NOTATION, 9  
NUMBERS, 4  
NUMBER, 4  
PUBLIC, 15, 16  
SDATA, 4  
SYSTEM, 16  
#IMPLIED, 10  
amp, 6  
apos, 6  
cdata, 9  
content, 12  
default, 4, 6  
element, 12  
end, 15  
entity, 9  
false, 8  
file, 12  
fixed, 4  
gt, 6  
idref, 9  
id, 9  
lt, 6  
name, 9  
nmtoken, 9  
notation, 9  
number, 9  
nutoken, 9  
on\_begin, 11, 12  
preserve, 6  
quot, 6  
rdf-syntax, 7  
rdf, 7  
remove, 6  
sgml, 5, 6, 11  
start, 15  
true, 8  
urlns, 7  
xmlns, 5, 7, 11, 13  
xml, 5, 6, 11  
catalog, 15  
declared, 16  
declared content). jteritemij, 9  
dialect, 7  
doctype, 8  
dtd/2, 8  
dtd-property/2, 8  
free\_dtd/1, 8  
free\_sgml\_parser/1, 11  
get\_sgml\_parser/2, 11  
implicit, 10  
load\_dtd/2, 8  
load\_html\_file/2, 3  
load\_sgml\_file/2, 3  
load\_structure/3, 4  
load\_xml\_file/2, 3  
new\_dtd/2, 8  
new\_sgml\_parser/2, 10  
open\_dtd/3, 8  
set\_sgml\_parser/2, 11  
sgml-parse/2, 11  
sgml\_register\_catalog\_file/2, 15  
xml2pl, 2